BANGLADESH UNIVERSITY OF ENGINEERING AND TECHNOLOGY
Electrical Circuits II Laboratory (EEE 106)
Simulation Experiment No. 01
Part -A: Transient Analysis of AC Circuits

Introduction:

A transient analysis deals with the behavior of an electric circuit as a function of time. When energized with a sinusoidal ac input, a circuit shows sinusoidal ac output in steady-state condition. However, before reaching steady-state condition a circuit passes through a transition period when the circuit is switched with an ac supply. During this transient period, the currents and voltages are not periodic functions of time. If a circuit contains an energy storage element(s), a transient can also occur in a dc circuit after a sudden change due to switching.

SPICE allows simulating transient behaviors by assigning initial conditions to circuit elements, generating sources, and the opening and closing of switches. Students are advised to apply the techniques for transient analysis of simple circuit laws and to verify the SPICE results by hand calculations.

Theory:

If an **RL** circuit is energized with an ac voltage then the expression of dynamic Equilibrium is:

$$L\frac{di}{dt} + Ri = E_m \sin(\omega t + \lambda)$$

where, i is the current through RL branch, $E_m \sin(\omega t + \lambda)$ is the applied voltage with λ phase.

The solution for current in RL circuit is:

$$i = \frac{E_m}{Z}\sin(\omega t + \lambda - \theta) - \frac{E_m}{Z}\sin(\lambda - \theta)\exp(-\frac{Rt}{L}) \quad \text{......................................(1)}$$

where, impedance, $Z = \sqrt{R^2 + (\omega L)^2}$ and phase difference between voltage and current, $\theta = \tan^{-1}(\frac{R}{\omega L})$.

In (1) first and second terms are steady-state and transient respectively. Mathematically, the steady-state term is known as the 'particular integral' and the transient term as the 'complimentary function'.

Similarly, for **RC** circuit the expression of dynamic Equilibrium is:

$$Ri + \frac{q}{C} = E_m \sin(\omega t + \lambda)$$

where, q is the charge and $i = \frac{dq}{dt}$

The solution for current in RC circuit is:

$$i = \frac{E_m}{Z}\sin(\omega t + \lambda + \theta) - \frac{E_m}{\omega RCZ}\cos(\lambda + \theta)\exp(-\frac{t}{RC}) \quad \text{.................................(2)}$$

where, impedance, $Z = \sqrt{R^2 + (1/\omega C)^2}$ and phase difference between voltage and current, $\theta = \tan^{-1}(-R\omega C)$.
In (2) first and second terms are steady-state and transient respectively.

For **RLC** branch circuit the expression of dynamic Equilibrium is:

$$L\frac{d^2 i}{dt^2} + R\frac{di}{dt} + \frac{1}{C}\frac{dq}{dt} = E_m \sin(\omega t + \lambda)$$

If $\frac{R^2}{4L^2} > \frac{1}{LC}$ then expression of current can be written as follow:

$$i = \frac{E_m}{Z}\sin(\omega t + \lambda - \theta) + \frac{E_d}{bL}\exp(-at)\sinh bt - \frac{E_m}{Z}\sin(\lambda - \theta)\exp(-at)\cosh bt$$

If $\frac{R^2}{4L^2} < \frac{1}{LC}$ then expression of current can be written as follow:

$$i = \frac{E_m}{Z}\sin(\omega t + \lambda - \theta) + \frac{E_d}{\beta L}\exp(-at)\sin \beta t - \frac{E_m}{Z}\sin(\lambda - \theta)\exp(-at)\cos \beta t$$

where,

$$E_d = E_m \sin\lambda - \frac{Qo}{C} - \frac{E_m \omega L}{Z}\cos(\lambda - \theta) - \frac{E_m R}{2Z}\sin(\lambda - \theta),\ a = \frac{R}{2L},\ b = \sqrt{\frac{R^2}{4L^2} - \frac{1}{LC}},$$

$\beta = -jb$ and Q_o is charge in capacitor before switching.

Report

Practice problem 1:



Figure 7.1

Construct the circuit in both netlist and schematic.
Using transient analysis, plot i, V_L by with V_m = 300 volts, frequency=50 Hz and lamda λ =0, 30, 60 degree.

Practice Problem 2:



Figure 7.2

Construct the circuit in netlist and using transient analysis, plot *i*, V_c vs time.

Practice Problem 3:



Figure 7.3

Construct the circuit in schematics and using transient analysis, plot *i*, V_L, V_C vs time.

Pre-lab work:

(1) Students will plot *i*, V_C, *and* V_L by using transient analysis and that will be verified during lab.
(2) They will calculate the maximum and minimum peak values of currents and voltages at transient condition and time required to reach steady state.
(3) For two different values of L, C, R measure maximum and minimum peak value at transient condition and time required to reach steady state.

Reference:

Alternating-Current Circuits – Russell M. Kerchner & George F. Corcoran

BANGLADESH UNIVERSITY OF ENGINEERING AND TECHNOLOGY
Electrical Circuits II Laboratory (EEE 106)
Simulation Experiment No. 01
Part -B: Analysis of Magnetically Coupled AC Circuits

Introduction:

If a one portion of the magnetic flux established by one circuit interlinks with a second circuit, two circuits are coupled magnetically and energy may be transferred from one circuit to the other through magnetic field. Where the windings are widely separated or are situated in space, the coupling is said to be loose. With closer proximity and proper orientations of the windings, coefficient of coupling approaches unity.



Figure Magnetic coupling with four component fluxes

The coefficient of coupling is defined as,

$$k = \sqrt{\left(\frac{\emptyset_{12}}{\emptyset_1}\right)\left(\frac{\emptyset_{21}}{\emptyset_2}\right)}$$

Mutual Inductances in PSPICE

Users of PSPICE often need to model inductors that are magnetically coupled. This may occur in steady-state power system simulations, or in power electronic transient circuit simulations where linear or nonlinear transformer models are used. In some cases it is necessary to model weakly coupled inductors. This tutorial will address the issues of modeling magnetic coupling in these circumstances.

Basic Linear Coupled Inductors



In the above figure two inductors are coupled by a coefficient of coupling, k. Their nodes are designated by small integers, and polarity marks have been added. The polarity information is passed to PSPICE by the order of the nodes. If the coefficient of coupling is k = 0.8, a valid PSPICE coding could be:

```
*name node1 node2 inductance (comment line)
L1    1     2     40mH
L2    3     4     10mH
*name ind1  ind2  k (comment line)
K12   L1    L2    0.8
```

Note that the polarity marks in the figure are beside nodes 1 and 3 of the inductors. In the listing, these are entered as the leftmost nodes. An equivalent polarity relationship could be indicated by reversing both nodes on both inductors. The coupling of the coils is entered by including a new part that must begin with the letter, K. The "K" part name is followed by a list of the coupled inductors, then by the value of the coefficient of coupling. The coefficient of coupling must occupy the range, $0 \le k \le 1$. Numerically the coefficient of coupling in practical installations may range from approximately 0.01 between certain types of radio circuits to as high as 0.98 or 0.99 between iron-core transformer windings.

Multiple Couplings with Different Values




```
La    1    2    15mH
Lb    3    4    12mH
Lc    5    6    10mH
Kab   La   Lb   0.08
Kbc   Lb   Lc   0.075
Kca   Lc   La   0.04
```

In the above figure, each inductor has mutual coupling with more than one other inductor, but with different coupling coefficient. In this case, PSpice requires a separate "K" part for each coefficient of coupling as shown in the code.

Note that the polarities of the inductors, and therefore the sense of the mutual coupling is accounted for by the order of the nodes entered for the self inductance parts. In this case, different symbols have been used in the figure to assure that it is understood which pairs are coupled and in what sense. In general, if there are n coils, there will be ½ n(n-1) "K" parts needed.

Multiple Couplings with Same Values




```
La    1    2     25uH
Lb    3    4     50uH
Lc    5    6    100uH
Ld    7    8    200uH
Kall  La   Lb   Lc   Ld   0.98
```

In the above example, we assume that all inductors share identical coefficients of coupling. This is a reasonable assumption when coil symmetry exists and all coils are wound on a common core. Under these conditions, PSpice allows a single "K" part to describe all the coupling.

Again, the polarity information is entered by the order of the nodes for the self inductances. Since all the coupling coefficients were the same, only one "K" part was needed instead of six.

Practice Problem:



FIGURE 5.1

a) A circuit with two mutually coupled inductors is shown in fig 5.1. Write down the netlist of the above circuit. If the input voltage is 120 V peak with a suitable frequency (say 60Hz, 90Hz etc). Calculate the magnitude and phase of the output current. The coefficient of coupling for the transformer is 0.999.
b) Observe the voltage waveshapes at point 2 and 4.
c) Perform the above problem by changing the position of the DOT on the Secondary inductor.
d) Now construct the circuit in PSPICE. Choose part “K_Linear” to establish a coupling between L1 and L2. Repeat steps 2 and 3 for the designed circuit.

Reference:

Alternating-Current Circuits – Russell M. Kerchner & George F. Corcoran

Bangladesh University of Engineering and Technology
Electric Circuits II Laboratory (EEE 106)
Simulation Experiment No. 02

Steady-State AC analysis, Frequency response and Filters

Pre Lab

- Read the Lab Sheet.
- Go to Figure 1. Use pen and paper to calculate the Voltages and Currents of all the nodes at 60Hz. Also, draw the phasor diagram.
- Go to Figure 2. Use pen and paper to find the expression of voltage (magnitude and phase) at node 2 in terms of frequency (f, where $f = 2 \times \pi \times f$).
- Determine the cut off frequency of the filter shown in Figure 2: equate the expression of the voltage at node 2 to $1/\sqrt{2}$ and find the corresponding frequency.

Introduction

In addition to DC circuit analysis, DC transient analysis, and AC transient analysis PSpice can be used to work on AC steady-state analysis.

AC Voltage and Current Sources

The syntax for an AC source is very similar to its DC counterpart. The AC source is assumed to be a ***cosine*** waveform at a specified phase angle. Its frequency must be defined in a separate " ***.AC*** " command that defines the frequency for *all* the sources in the circuit. The unique information for the individual source is: the name, which must start with "V" or "I," the node numbers, the magnitude of the source, and its phase angle. Some examples follow.

```
*name   nodelist   type   rms value  phase(deg)
Vac     4    1     AC     120V       30
Vba     2    5     AC     240            ; phase angle 0 degree
Ix      3    6     AC     10.0A      -45 ; phase angle -45 degree
Isv     12   9     AC     25mA           ; 25 milliamps @ 0 degree
```

Here source type, AC, *must* be specified, because the default is DC. If the phase angle is not specified it will be assumed as zero degrees. The units of the phase angle will be in degrees. For voltage source the node on the left is the positive node and the node on the right is the negative node. Similarly, in case of current source positive current flows into the source from the node on the left, passes through the source, and leaves the source from the node on the right.

In some of your previous experiments you have used "**SIN**" type of source which is one of several useful source types (also EXP, PULSE, PWL, etc) that are used for ***transient* analysis**. Do not attempt to use *SIN* for steady-state (phasor) AC analysis nor for frequency sweeps. The

SIN type is a time-based function for time-based analysis, whereas the AC type is used in frequency-based modeling. Since phasor analysis uses frequency-based models of circuit elements, always use the AC type as described in this experiment.

Use of the .PRINT AC Command

To enable **.PRINT** command **.AC** command must be used. The .AC command was designed to make a sweep of many frequencies for a given circuit. This is called a *frequency response.* Three types of ranges are possible for the frequency sweep: LIN, DEC and OCT. At this time we only want a single frequency to be used so it does not matter which one we choose. We will pick the LIN (linear) range to designate our single frequency.

```
 *   type #points  start stop
.AC  LIN   1       60Hz  60Hz;  <== single frequency
.AC  LIN   6       100   200;   <== a linear range sweep
.AC  DEC   20      1Hz   10kHz; <== a logarithmic range sweep
```

The first statement above performs a single analysis using the frequency of 60 Hz. Placing the units "Hz" after the value is optional. The second statement would perform a frequency sweep using frequencies of 100Hz, 120Hz, 140Hz, 160Hz, 180Hz, and 200Hz. The third statement performs a logarithmic range sweep using 20 points per decade over a range of four decades. This will be useful later for studying frequency response of circuits.

Finally, we can discuss the actual .PRINT AC command. Printing the components of phasor values (complex numbers) requires some options. There are four expressions needed for this: magnitude, phase (angle), real part and imaginary part. In addition, we can print voltages or currents. For instance, to print the magnitude of a voltage between nodes 2 and 3, we would specify "VM(2,3)." The phase angle of this same voltage would be "VP(2,3)" and would be printed in degrees. If we need the current magnitude through resistor Rload, we would specify "IM(Rload)." The real part of the voltage on node 7 would be specified "VR(7)" and its imaginary part, "VI(7)." As with the .PRINT DC command, there is no limit on the number of times it can be used in a listing; nor is there a limit on how many print requests can be on a single line.

```
.PRINT AC VM(30,9) VP(30,9); magnitude & angle of voltage
.PRINT AC IR(Rx) II(Rx); real & imag. parts of current through Rx
.PRINT AC VM(17) VP(17) VR(17) VI(17); the whole works on node 17
```

Specifying frequency range for AC Sources

.AC command is used to specify one of the following three types of frequency ranges.

LIN (linear) Range Type

The LIN range type is linear. It divides up the range between the minimum and maximum user-specified frequencies into evenly spaced intervals. This is best used to view details over a narrow bandwidth. The first parameter after the keyword LIN is the number of points to calculate. This is followed by the lowest frequency value in Hz, then the highest frequency value in Hz. As with all the range types, the unit "Hz" is optional.

```
*   type #points  start stop
.AC LIN  101      2k    4k    ; 101 points from 2 kHz to 4 kHz
.AC LIN  11       800   1000 ; 11  points from 800 Hz to 1 kHz
```

OCT (octal) Range Type

The OCT range is logarithmic to the base two. Thus each octave has the same number of points calculated. This is somewhat useful for designing electronic equipment for musical applications. However, the resulting graphs are very similar in appearance to sweeps made with the DEC range. The first parameter after the keyword OCT is the number of points per octave to calculate. This is followed by the lowest frequency value in Hz, then the highest frequency value in Hz.

```
*   type #points  start stop
.AC OCT 20 440Hz 1.76kHz; 20 points/octave over 2 octaves
.AC OCT 40 110Hz 880Hz  ; 40 points/octave over 3 octaves
```

DEC (decade) Range Type

The DEC range is logarithmic to the base ten. Thus each decade has the same number of points calculated. This is the most commonly used range for making *Bode plots* of a frequency response. The first parameter after the keyword DEC is the number of points per decade to calculate. This is followed by the lowest frequency value in Hz, then the highest frequency value in Hz.

```
*   type #points  start stop
.AC DEC  50       1kHz  100kHz ; 50 points/decade over 2
                               ; decades
.AC DEC  25       100k  100MEG ; 25 points/decade over 3
                               ; decades
```

Probe

The independent variable used by PROBE in a .TRAN analysis is *time*. But in a frequency sweep the independent variable used by PROBE is *frequency*. When PROBE stores data in a transient (.TRAN) analysis, the dependent variables are instantaneous voltages and currents; whereas in a frequency sweep these dependent variables are real and imaginary components of phasor voltages and currents.

Example Circuit 1: Steady State AC Analysis

We will analyze the following circuit at a frequency at 60 Hz.



Figure 1. Circuit for Example 1

Corresponding Netlist of the Circuit

```
;60 Hz AC Circuit
Vs  1  0  AC  120V  0
Rg  1  2  0.5
Lg  2  3  3.183mH
Rm  3  4  16.0
Lm  4  0  31.83mH
Cx  3  0  132.8uF
.AC LIN 1 60 60
.PRINT AC VM(3) VP(3)
.PRINT AC IM(Rm) IP(Rm)
.PRINT AC IM(Cx) IP(Cx)
.END
```

Example Circuit 2: Frequency Sweep

We will look in to a simple circuit to understand AC Frequency Sweep.



Figure 2: Circuit for Example 2

Corresponding Netlist of the Circuit

```
;First-order low-pass RC filter
Vin 1   0 AC 1.0V
R1   1   2 0.25
C1   2   0 50uF
.AC DEC 20 100Hz 100kHz
.PROBE
.END
```

The circuit in Fig. 2 is a first-order low-pass filter. Since we want the gain of this filter, it is convenient to make the input voltage 1 volt so the output voltage is numerically equivalent to the gain. However, the post-processor within PROBE is fully capable of performing arithmetic such as dividing the input voltage into the output voltage.

Plotting In Terms of Voltage

After running this in PSpice, start PROBE, choose "Add" from the "Trace" menu and plot the output voltage. PROBE will provide the following graph.



Figure 3: Voltage across Capacitor for the Circuit in Fig. 2.

Plotting the Gain (in Terms of Decibel)

Another option is to have PROBE plot the gain in decibels. To do this, choose "Add" from the "Trace" menu in PROBE. Then select the "DB" function in the right-hand column and choose "V(2)" from the left-hand column. After selecting "OK," you should see the following trace.



Figure 4: Plot in terms of Decibel

Notice that the gain is -3db at a frequency of 1 kHz (the half-power frequency) and declines at 20 dB/decade thereafter.

Plotting the Phase

The remaining demonstration for this example is to have PROBE plot the *phase shift* of the low-pass filter as a function of frequency. We simply specify "VP(2)" from the "Add Trace" dialog box. Notice that this is the same format used in the .PRINT AC command in PSPICE. PROBE automatically shows the angles in degrees.



Figure 5: Plot of the Phase

Post Lab

Answer the following questions in your lab report. Your report should include necessary figures, plots and explanations.

1. In the example 2, we used a source voltage of $1\angle 0$. And plotted the gain in decibel just by plotting the voltage at node 2. But in real life the source voltage would not necessary be unity. It may as well be any arbitrary voltage $V\angle\theta$. How would you plot the gain (in decibels) in such case? Devise a method.

   Use $6\angle 0$ in the circuit of figure 2. Perform frequency Sweep and Plot the gain by the method you devised. At what decibel do you get the cutoff frequency this time? Does the cutoff frequency depend on source voltage?

2. Determine the cutoff frequency of the high-pass filter shown in figure 6 by pen and paper. Simulate the circuit using Frequency Sweep. Keep the cutoff frequency within the range of the sweep. Find the cutoff frequency from the plot of the voltage. Does the frequency obtained from the simulation and from the pen and paper match?



Figure 6: High Pass

3. Cascade the low-pass and high-pass filters shown in figure 2 and figure 6 and form a band pass filter.

   Then find the center frequency of the circuit using frequency sweep. Center frequency is the frequency at which the gain is highest.



Figure 7: Band Pass Filter.

   And then find the lower and upper cutoff frequencies. These frequencies are the frequencies at which the gain is -3db.

4. Now, we will explore the second order filters using RLC circuits. The following circuits are collected from the Wikipedia page of the RLC Circuits (https://en.wikipedia.org/wiki/RLC_circuit). You can look up in the page and find out what kind of filters are each of these. However, it will be beneficial if you perform the frequency sweep, look at the gain plot and determine what kind of circuits are these.

   As usual use $1\angle 0$ as the source voltage for each of these circuits. Perform, frequency sweep and determine what kind of filters are these. We are not necessary interested at the cutoff frequencies for the purpose of the report. Look up the Wikipedia page and see the equations of the cutoff frequencies. These are easy to derive following standard procedure.



Figure 8 (a)



Figure 8 (b)



Figure 8 (c)



Figure 8 (d)



Figure 8 (e)



Figure 8 (f)

   Use suitable values for R_L, L, and C for simulation.

5. There are 8 configurations in 4. There are few more configurations which can be designed. Can you design your very own second order filter? Perform a frequency sweep on the filter you designed. What kind of filter it is? What are the cut off frequency/frequencies and center frequency (if any)?

   (Optional: Find the expression of the output voltage for the filter you designed and comment on the characteristics of the filter based on the equation.)

Bangladesh University of Engineering and Technology
Electric Circuits II Laboratory (EEE 106)
Simulation Experiment No. 03

Series and Parallel Resonance & Power Factor Correction

Pre Lab

1) Read the lab sheet.
2) Bring the text book with you.
3) Read the series resonance article from your text book. You should know about the results for the three cases, i.e. when frequency, inductance or capacitance is varied.
4) Read the text on parallel resonance from the book. We will be dealing with the parallel RC and RL branch during the lab as in the book.

Theory

Resonance:

A circuit is in resonance when the source power factor is 1. That is the voltage and current are in phase.

Series Resonance:

A series circuit containing R, L and C is in resonance when the resultant reactance is zero. In such situations the current and voltage becomes in phase.



Figure 1: Series Resonance.

Mathematically, $X_L = X_C$

So solving the equation for f we get,

$$f = \frac{1}{2 \times \pi \times \sqrt{LC}} \dots (1)$$

So resonance in a series RLC circuit can be obtained by varying inductance, capacitance or operating frequency.

Parallel Resonance:

In parallel RL and RC branch, in order for the voltage and current to be in the same phase the sum of the admittance of the two branches (RL and RC) should yield only conductance. That is the sum of susceptance of the two branches should be zero.



Figure 2: Parallel Resonance

This condition provides a characteristic equation for the circuit:

$$\frac{2\pi fL}{R_L^2 + (2\pi fL)^2} = \frac{\frac{1}{2\pi fC}}{R_C^2 + \left(\frac{1}{2\pi fC}\right)^2} \quad \dots (2)$$

Writing them in terms of impedance:

$$\frac{X_L}{R_L^2 + X_L^2} = \frac{X_C}{R_C^2 + X_C^2} = G$$

Then solving for X_L,

$$X_L = \frac{1 \pm \sqrt{1 - 4G^2R_L^2}}{2G}$$

So, valid solutions (positive and real numbers) from this equation can be obtained when,

$$4G^2R_L^2 \leq 1$$

So, for a given$f$,R_L, R_C, and C we can calculate L from the characteristic equation, when the inequality provided above is satisfied. (Can resonance be obtained when the inequality is not satisfied?)

Now, similar conditions can be obtained for any of the parameters given other four parameters have been fixed. These are not stated here, but students should explore on their own.

Power Factor Correction:

Power factor correction is the addition of a reactive element (usually a capacitor, as most domestic and industrial loads are inductive) in parallel with the load in order to make the power factor closer to unity. Its concept is similar to parallel resonance described above.

Since most loads are inductive, a load's power factor is improved or corrected by deliberately installing a capacitor in parallel with the load. The effect of adding the capacitor can be illustrated using either the power triangle or the phasor diagram of the currents involved.

The circuit in Fig. 3(a) has a power factor of cos θ_1, while the one in Fig. 3(b) has a power factor of cos θ_2. It is evident from Fig. 3(c) that adding the capacitor has caused the phase angle between the supplied voltage and current to reduce from θ_1 to θ_2, thereby increasing the power factor. From the magnitudes of the vectors in Fig. 3(c) it is also evident that with the same supplied voltage, the circuit in Fig. 3(a) draws larger current $\boldsymbol{I_L}$than the current $\boldsymbol{I}$drawn by the circuit in Fig. 3(b). Power companies charge more for larger currents, because they result in increased power losses (by a squared factor, since $P = I_L^2R$). Therefore, it is beneficial to both the power company and the consumer that every effort is made to minimize current level or keep the power factor as close to unity as possible. By choosing a suitable size for the capacitor, the current can be made to be completely in phase with the voltage, implying unity power factor.



Figure 3: Power factor correction (a) original inductive load (b) inductive load with a compensating capacitor in parallel (c) Phasor Diagram showing the effect of inserting shunt capacitor



Figure 4: Power triangle illustrating power factor correction

If the original inductive load has apparent power S_1, then

$$P = S_1 cos\theta_1$$

$$Q_1 = S_1 sin\theta_1 = P tan\theta_1$$

If we desire to increase the power factor from cos θ_1 to cos θ_2 without altering the real power, then the new reactive power is

$$Q_2 = P tan\theta_2$$

The reduction in the reactive power caused by the compensating element is given by,

$$Q_{comp} = (Q_1 - Q_2) = P\ (tan\theta_1 - tan\theta_2)$$

If the original load is inductive, the compensator is to be a capacitor and its value can be determined as follows.

$$Q_C = \frac{V^2}{X_C} = \omega V^2 C$$

$$C = \frac{Q_C}{\omega V^2} = \frac{P\ (tan\theta_1 - tan\theta_2)}{\omega V^2}$$

However, if the original load is capacitive, the compensator is to be an inductor and its value can be determined as follows.

$$Q_L = \frac{V^2}{X_L} = \frac{V^2}{\omega L}$$

$$L = \frac{V^2}{\omega Q_L} = \frac{V^2}{\omega P\ (tan\theta_1 - tan\theta_2)}$$

The ability to adjust the power factor of one or more loads in a power system can significantly affect the operating efficiency of the power system. The lower the power factor of a system, the greater the losses in the power lines feeding it. Most loads on a typical power system are induction motors, so power systems are almost invariably lagging in power factor. Having one or more leading loads (overexcited synchronous motors) on the system can be useful for the following reasons:

- A leading load can supply some reactive power Q for nearby lagging loads, instead of it coming from the generator. Since the reactive power does not have to travel over the long and fairly high-resistance transmission lines, the transmission line current is reduced and the power system losses are much lower.
- Since the transmission lines carry less current, they can be smaller for a given rated power flow. A lower equipment current rating reduces the cost of a power system significantly.

Frequency Analysis

From Analysis Setup > AC Sweep. Then you should be able to choose the frequency sweep characteristics.



Parametric Analysis (Review)

1) Parametric analysis allows you to run another type of analysis (transient, sweeps) while using a range of component values using the **global parameter** setting. The best way to demonstrate this is with an example, we will use an inductor, but any other standard part would work just as well (capacitor, resistor).

2) First, double-click the value label of the Inductor that is to be varied. This will open a "Set Attribute Value" dialog box. Enter the name **{RVar}** (including the curly braces) in place of the component value. This indicates to PSpice that the value of the inductor is a global parameter called RVAL. In order to define the RVAL parameter it is necessary to place a global parameter list somewhere on the schematic page. To do this, choose "**Get New Part**" from the menu and select the part named **param.**



3) Place the box anywhere on the schematic page. Now double-click on the word **PARAMETERS** in the box title to bring up the parameter dialog box. Set the NAME1= value to **RVar** (no curly braces) and the VALUE1= value to the nominal inductance value (10mH for example). This nominal value is required, but it is only used if the DC bias point detail is computed. Otherwise, the value is ignored by PSpice.



4) Finally, go to the "Analysis Setup" menu and enable "Parametric" analysis. Open the Parametric setup dialog box and enter the sweep parameters: Name: **RVar** Swept variable type: Global Parameter. Make sure the other analysis type(s) are selected in the analysis setup menu (transient, sweeps). PSpice will now automatically perform the simulation over and over, using a new value for **Rvar** during each run.

Example Circuit 1

1) Draw the circuit in Schematics. Voltage source is VAC with 120V RMS and 0 degree phase.



2) Use equation (1) to find the resonant frequency of the circuit.
3) Choose a suitable frequency range and perform frequency sweep. Choose a large frequency range in order to get an idea of the asymptotic behavior of the circuit in terms of frequency.
4) Plot magnitude of the current. The things you should notice
   a) The value of current is max at resonant frequency.
   b) From resonant frequency if you move in either direction the current is decreasing.
5) Add a plot to window. Plot the phase of the current. You should notice:
   a) The value of phase at resonant frequency.
   b) The limiting value of phase when you are going left from the resonant frequency.
   c) The limiting value of phase when you are going right from the resonant frequency.
6) Now, in a new probe window, plot the following quantities in terms of frequency in the same plot,
   a) V_L vs.f
   b) V_C vs.f
   c) V_R vs.f
   d) I vs.f

   The scale of the I vs f curve may be very low. In that case multiply the value of I by a constant.

   You should notice the following:
   a) The curves should have the same shape as you have seen in the book.
   b) Maximum of the inductor voltage occurs after resonant frequency as explained in the theory.
   c) Maximum of the capacitor voltage occurs before resonant frequency.
   d) At resonant frequency the inductor and capacitor have the same voltage.

Example Circuit 2

1) Draw the circuit in schematics.



2) We want to find the resonance at $60Hz$. Using equation (1) find the value of the inductance which causes resonance.
3) Using the obtained value perform a parameter sweep on L at $60Hz$ that is use linear frequency sweep at that frequency. Use reasonably large range.
4) In the probe window plot the following curves in the same plot:
   a) V_L vs. L
   b) V_C vs. L
   c) V_R vs. L
   d) I vs. L

   As previously, if shape of any of the curve cannot be understood from the plot, use necessary scaling.

Example Circuit 3



1) The following circuit has been constructed in such a way that resonance is obtained at $60Hz$ frequency. Verify this from equation 2.

2) Draw the circuit in schematic and perform a frequency sweep in a suitable frequency range (1Hz to 100Hz for example).
3) You should observe the following:
   a) The phase of the current I(R3) is zero at resonance frequency.
   b) The maximum impedance and consequently minimum current is not at resonant frequency.
4) You should be able to explain the observations you made in (3).

Example Circuit 4

1) This circuit is a common form of transmission line in single phase. The Resistance R1 is the line resistance and the RL branch may constitute a electric machine. And the capacitor C1 is the capacitor used for power factor correction.



2) In this example we are interested at 50Hz frequency.
3) First draw the circuit without the capacitor C1. Perform a steady state analysis at 50Hz and calculate the power absorbed in the resistance R1. Note down this value.
4) Now add the capacitor C1 and perform a parametric sweep on the capacitor in suitable range (for example, 1uF to 1mF) so that resonance is obtained at 50Hz.
5) Now calculate the power absorbed in resistance R1 near the resonant frequencies. The following observations should be made:
   a) At resonance the power absorbed in resistance R1 is lower than that was observed in (3).
   b) The capacitance at which the absorbed power is lowest is not the capacitance at which resonance is obtained.
6) Explain your observations.

Post Lab

Along with the lab works include these in your lab report including necessary explanations and figures.

1) In the circuit of "Example Circuit 2" use L=25mH. Then perform a parametric sweep on the Capacitance and find the capacitance at which resonance is obtained. Does this match with the value obtained from Eq. (1). Also, plot the following graphs:
    a. V_L vs. C
    b. V_C vs. C
    c. V_R vs. C
    d. I vs. C
2) Find the equation of the following circuit for the resonant condition:



For a fixed value of f, R_L, L, and R_C find the condition(s), if any, that will guarantee valid solution(s) for C from that equation.

Bangladesh University of Engineering and Technology
Electrical Circuits II Laboratory (EEE106)
Simulation Experiment no 4

Sub-Circuits using Net listing and Schematics and Three Phase Circuits

Objective:

The objective of this experiment is to learn how to write and invoke a sub-circuit and implement the concept in three phase circuits.

Simple Subcircuits in PSpice

One of the more useful concepts in PSpice is the use of *subcircuits* to group elements into clusters in order to replicate the clusters without having to re-enter all the elements each time. This is very useful for several reasons. First is the labor savings of replacing many lines of circuit data with a single subcircuit call. This also reduces the chance of making a typo. Second, the use of a subcircuit usually improves clarity by removing confusing clutter. The user can suppress printing unwanted details internal to a subcircuit, thus making the output easier to understand. If desired, the user can place often-used subcircuits into an *include* file so that the main source file for the problem is kept simple. Then the definition of the subcircuit is out of sight entirely.

Coding a Subcircuit

Each subcircuit used in a study must have a unique name. This is true of any other circuit element. Also, there must be a list of at least two nodes that can be connected to elements external to the subcircuit. A subcircuit can have many external node connections, if needed. Later, we will find that parameters can be passed to a subcircuit in order to allow unique behavior and responses from an instance of a subcircuit.

The initial line of a subcircuit section must begin with ".SUBCKT," followed by the name and then the external node list. After that, optional features (not to be discussed yet) can be added. The best method of understanding the use of a subcircuit is by example. Below, we find a cluster of components that can be combined into a subcircuit.



Fig. 1 : Subcircuit of Example 1

Note that nodes 5, 12 and 18 have external connections. Therefore, they must be included in the node list in the subcircuit definition. Nodes 10 and 13 do not have external connections and need not be (indeed *should* not be) included in this node list. They are internal nodes and will be used to help define the subcircuit. Now, we can code the above subcircuit as follows. Note that the code could be embedded into the rest of the code for the main circuit or could be placed in a separate *include* file.

```
*        name          nodelist
.SUBCKT Example_1   5   12   18
Iw    10    12    DC    10A
Ra     5    12    5.0
Rb     5    13    4.0
Rc    12    13    2.0
Rd     5    18    8.0
Re    13    18    3.0
Rf    10    13    1.0
Rg    10    18    6.0
.ENDS
```

Note that the subcircuit section must be terminated with a ".ENDS" command.

Invoking a Subcircuit

All subcircuit calls are made by declaring a part with a unique name beginning with "X," followed by the node list and then the subcircuit name. The node list in the calling statement must have the same number of nodes as the node list in the subcircuit definition. To demonstrate the use of the calling statement, we present the following main circuit which contains two instances of the above subcircuit. X1 and X2 are the two instances of the subcircuit "Example_1." For added clarity, the subcircuit's defined external nodes are shown in parentheses. Note that these nodes are mapped into the main circuit by *different numbers.*



Fig. 2: Circuit using the subcircuit Example_1

```
Subcircuit Example No. 1
*          name            nodelist
.SUBCKT Example_1    5    12    18
Iw    10    12    DC    10A
Ra     5    12    5.0
Rb     5    13    4.0
Rc    12    13    2.0
Rd     5    18    8.0
Re    13    18    3.0
Rf    10    13    1.0
Rg    10    18    6.0
.ENDS

Vs     1     0    DC    50V
Ra     1     2    1.0   ; different from Ra above
Rb     3     4    3.0   ; different from Rb above
```

```
Rc      7     0   25.0   ; different from Rc above
Rd      6     0   45.0   ; different from Rd above

Iq      0     5    DC    15A
*       nodelist         name
X1      2     7     3    Example_1
X2      4     6     5    Example_1
.END
```

Subcircuits using Schematics:

DCLICKL=Double CLICK Left, CLICKL=CLICK Left

<u>**Step 1:** Create the MODEL</u>

1. Draw RLC series circuit with R=100Ω, L=0.31H and C=31.83µF.
2. Replace the input and output nodes with the *interface port symbols* IF_IN and IF_OUT (Search IF_IN and IF_OUT from the part list). To label each port symbol, **DCLICKL** on each symbol and fill in with any name you want.
3. Save the schematic to file *Ex_1.sch*.
4. Go to **Tools, Create Subcircuit** to generate the subcircuit MODEL definition. This model definition is automatically stored as *Ex_1.sub* in the same directory as *Ex_1.sch*.
5. To make the MODEL definition (*Ex_1.sub*) available locally (valid only to *Ex_1.sch*): Go to **Analysis, -> Library and Include Files**, enter *Ex_1.sub* in the File Name Field, **Add Library** (no asterisk), **OK**. (Add Library* would make the model global and available to all schematics).



Step 2: Create the PART

6. To create a PART, we first create a symbol. Assign the symbol a PART name (such as Ex_1), and store the PART in a symbol library (such as *userlib.slb*). The procedure to do this is as follows:

File, Symbolize, type PART name *Ex_1* in *Save As* dialog box, **OK**



After clicking OK the ***Choose Library for Schematic Symbol*** *dialog box will appear*.



To save part *Ex_1 in* library *userlib.slb* in a specific directory **CLICKL** to select *userlib*, Then click **OPEN**.

NOTE: If no .slb file appears you can write any name in the File name box i.e "uselib.slb"/ "anyname.slb". Then click OPEN. A .slb file will be created automatically

7. The next step is to edit the new PART symbol by bringing up the Symbol Editor: Click **File,**Then **Edit Library.** A new Window will open. Then in the new window click **File, Open, DCLICKL** on *userlib.slb*,

Part, Get, DCLICKL on *Ex_1* to bring up the initial symbol.



8. Modify the initial symbol by moving the pins and attributes by **select, DRAG**. Also, **DCLICKL** on any attribute or pin to place pin numbers and reorient pin names. You may wish to remain with the initial symbol as well.
9. To save the edited PART symbol again in the symbol library (*userlib.slb*): **File, Save** (and if necessary, **Yes** add to list of configured libraries).

Step 3: Associate the MODEL with the PART

10. To associate the MODEL definition (*Ex_1.sub*) with the PART symbol (in *userlib.slb*): **Part, Attributes, CLICKL** on PART=, enter *Ex_1* in Value field, **Save Attr, CLICKL** on MODEL=, enter *Ex_1* in Value field, **Save Attr, OK**. (**DRAG** part name *Ex_1* to desired position.)



11. To return to Schematics, **File, Close,** (and if necessary, **Yes** save changes to Part, **Yes** save changes to library).
12. **File, Close** to exit the *Ex_1.sch* window (and if necessary, **Yes** save all changes).

Circuit design using the subcircuit

13. Create a new schematic screen.
14. Now If you search in the Part_list you will find "Ex_1" as a part .

Exercise 1:

Use the subcircuit in three different phases of a balanced three phase supply with f=50Hz and amplitude=100V. Connect the sources and the loads in

a)YY,

b)YΔ,

c) ΔY

d)ΔΔ

Use Line impedances of 1+ j20 **Ω**. Use another subcircuit for the line impedance .

For each case Plot

a) Line voltage and phase voltage in the same window. What is the relationship between the two voltages?
b) Line current and phase current in another window .What is the relationship between the two currents?
c) Plot the voltage and current of the load in the same plot. Find the power factor of the load from the zero crossings of voltage and current waveform. (Use proper scaling if necessary)

Note: While working with delta-connected sources, you have follow the instructions otherwise Pspice will show errors.

1. insert a resistor of negligible resistance (say, 1 $\mu\Omega$ per phase) into each phase of the delta-connected source.
2. insert balanced wye-connected large resistors (say, 1MΩ per phase) in the delta-connected source so that the neutral node of the wye-connected resistors serves as the ground node 0.

Post Lab:

1. **Unbalanced Load**



assume that the three-phase circuits are unbalanced and operating in the positive sequence with Va= 170 $\angle 0$ V. Use the line impedance (1 +j10)
impedance, but the load is now (20 + j20) Ω for phase A, (50 + j10) Ω for phase B, and (5 + j50) for phase c. Find

a) the line currents Ia Ib Ic and the neutral current (In)in peak values
b) the power loss in each line, including the neutral
c) the power factor for each phase of the load.
d) Complex, Real and reactive power absorbed by load

Does the answer match with your calculation?

2. **Power Factor correction**

Now suppose that the loads from the previous postlab task are balanced ,with impedance of 20+j20 Find

a) the power factor for each phase of the load.
b) From the circuit provided below find the kVAR rating of the three capacitors delta connected in parallel with the load that will raise the power factor to 0.9 lagging.

3. **Balanced and Unbalanced load together**

   When you are trying the find the magnitude and phase of any signal you can also use IPRINT and VPRINT parts from the part list. Using these parts from the part list complete the following task.

   A 3-phase line has an impedance of 3 + j4 Ω. The line feeds two balanced three-phase loads that are connected in parallel. The first load is Y-connected and has an impedance of 30 + j40 Ω/phase. The second load is delta connected and has an impedance of 60 - j45 Ω/phase. The line is energized at the sending-end from a 3-phase balanced supply of line to neutral voltage $V_{an} = 200\angle 0^\circ$ V (rms), 60 Hz. Determine

(a) Current in the line for each phase.

(b) Current in each phase of the Y-connected loads.

(c) Current in each phase of the delta connected loads.

.